Profit Optimization in a Catfish Resale Firm

Bus Adm 713 Final Project

Michael Bryant and Jeffrey Grieger
5-12-2014

Profit Optimization in a Catfish Resale Firm

Executive Summary

A catfish resale firm which handles processing and packaging prior to resale needs to optimize their profit. The firm processes three different grades of fish and sells it to five groups of customers. The fish is sold at a set price per pound regardless of grade, but profit varies by each grade to do processing time and labor costs. The firm needs a production schedule and assignments for each employee to ensure that demand is met for each customer while profit is maximized.

A multi-tiered casual analysis model is used to optimize the profit by creating the optimal production schedule. Future demand was determined with an ARIMA model in SAS using historical data of sales of catfish in pounds and price. A forecast lead of three months is necessary so that that orders can be placed with the catfish farms in time to meet future demand. Price per pound will be set at 322 cents in three months. The forecasted demand is then used in the MTCA along with each employees time (percentage based) spend on processing each grade of fish and the maximum output in pounds the employee can process during the time period. Revenue per pound of each grade of fish, per customer is also a factor in the MTCA.

Future demand based on the set price of 322 cents per pound of catfish was determined to be 11860.26 lbs. The firm can currently meet this demand with the four employees on staff. These employees have varying skill sets and process more than one grade of fish. The results of the MTCA determine that demand can be met with an optimized profit by only having Employees 1 and 4 process fish. Employee 1 specializes in the boneless and skinned grade, while Employee 4 can meet the demand for whole fish packaging and ground/processed fish. Prior to making a decision to reduce the workforce by half, the recommendation to the firm is to analyze other firm controlled variables such as pricing structure. Setting a different price point for each grade and customer contract may increase revenue to such a level that more than two processors would be needed to maximize profit.

Introduction:

This goal of this project is to simulate product demand and develop an optimized production schedule based on this forecasted demand. To do this, a multi-tiered causal analysis (MTCA) type model is created, as the forecasted demand values are placed into the supply model. In this scenario, a catfish processing and packaging company will maximize the profit on their catfish products by first forecasting demand three months in advance and then using an optimization program to optimize a production schedule in that third future month.

The forecasting equation in this simulation is an ARIMA model that will model the dependent variable of sales(per lbs) on the independent variable price(cents per lbs), which is under direct management control and future values are known. The forecast of three months is used because catfish farms need approximately three months to grow an acceptable catfish of an industry standard two pounds (http://www.catfish.net/basics.html). This will allow the company enough time to inform its suppliers of the expected change in demand based on managerial a decision to raise the price of their catfish to 322 cents per lbs., a 16.4% increase in price from the current month of February 2013.

This simulation assumes the date is March 1^{st} 2013, as the data collected from the http://www.ers.usda.gov/data-products/aquaculture-data.aspx is from January 1986 to February 2013. This simulation also assumes only five customer types; High end Restaurant, Middle Tier Restaurant, Fast Food Restaurant, and Pet Food company. Also assumed is the percent consumption of the fish products by each customer, which come in three types; whole catfish, skinned and boneless catfish, and ground and processed catfish. In addition, production labor hour per catfish product is also known. And finally, revenue per customer by catfish type is known. By having this data, along with the forecasted demand, an MTCA model can be created to optimize profit three months in the future when the price change will occur.

Using SAS Enterprise Guide software, an ARIMA model using Proc Arima is developed with forecasts. Also, using SAS Enterprise Guide software, an optimization program is created using Proc optmodel from the simulated data. The details of these models, that together make the MTCA model, will be presented in the proceeding sections.

I. Demand Model and Forecast

First and foremost, the demand function must be calculated in an MTCA. For this particular problem, an ARIMA was decided upon to model demand. The independent variable is price(cents per lbs), which is under management control, and the dependent variable is sales of catfish (lbs). It should be noted that this is the data for all of the United States. For the sake of ease, it is assumed this data is the same as our target market and represents the whole of our customers. The full code is in the index, only relevant code and output will be displayed in the body of this paper.

The first step in analyzing the data is to determine if our variables need any differencing or scale transformation, or if there are any seasonal trends. By simply modeling our sales data (figure 1a), we can see there are appears to be some seasonality with peaks and troughs in evenly spaced intervals. Because we are using an ARIMA model, it is also best to make the sales and price data stationary, as it can be seen that sales is not stationary, this is accomplished by differencing the data.

Figure 1a. Plot of Raw Sales Demand



After differencing, the data for the PACF and ACF plots (figure 1b) suggest we need to add auto-regressive and moving average terms. Also, we see a clear pattern with large lags every twelve months, which is consistent with the figure 1a of sales that implies seasonality.

Figure 1b. Correlation Analysis for Sales Difference Sales Data



After trial and error with ARIMA model, it was found the best model had p=q=1 and also p=q=12, the 12 representing the seasonality affect which is both an autoregressive and moving average seasonality. The following code produced the best ARIMA model that was attempted:

```
proc arima data=comb out=outarima;
identify var=Sales(1) crosscorr=Price(1) nlag=36;
estimate input=Price p=(1)(12) q=(1)(12) method=ml noconstant;
forecast lead=3 ID=Date interval=month;
run;
```

No constant was used because it is assumed after differencing the average is zero. The following output summarizes the ARIMA model and residual diagnostics.

Table 1a: Coefficients of the ARIMA Model

Maximum Likelihood Estimation							
Parameter	Estimate	Standard Error	t Value	Approx Pr > |t|	Lag	Variable	Shift
MA1,1	0.74871	0.05692	13.15	<.0001	1	Sales	0
MA2,1	0.74863	0.04851	15.43	<.0001	12	Sales	0
AR1,1	0.18652	0.08267	2.26	0.0241	1	Sales	0
AR2,1	0.98808	0.0073040	135.28	<.0001	12	Sales	0
NUM1	-48.76064	5.20361	-9.37	<.0001	0	Price	0

The back-shift notation of the model can be seen in the following output:

Model for variable Sales	
Period(s) of Differencing	1

No mean term in this model.

Autoregressive Factors	
Factor 1:	1 - 0.18652 B**(1)
Factor 2:	1 - 0.98808 B**(12)

Moving Average Factors	
Factor 1:	1 - 0.74871 B**(1)
Factor 2:	1 - 0.74863 B**(12)

Input Number 1	
Input Variable	Price
Period(s) of Differencing	1
Overall Regression Factor	-48.7606

Figure 1C: Correlation Analysis for ARIMA model Residuals



The PACF, AFC, IACF produce ideal plots. The above information shows that this model is a good model for the data.

There are some weaknesses in the demand model. First, it should be noted that while other models such as exponential smoothing and regular regression were attempted, a thorough analysis of each of these other models was not executed. It could also be the case that a combined forecast would be better as well. Secondly, the ARIMA model was not executed with an out of sample data set to check the accuracy of the demand. The primary goal of this simulation is to run an optimization model, and therefore it is assumed that the ARIMA model is accurate based on that test.

Using the forecast lead=3 (months), as explained in the introduction, was used to produce sales/demand data for three months in the future.

Management will raise price in three months' time to 322 cents per pound. The price for April and March of 2013 will remain the same as February 2013. Therefore our forecasted demand (forecast) data for those three future months is displayed in table 1b.

Table 1b: Forecast Sales Data

Obs	DATE	Sales	FORECAST	STD	L95	U95	RESIDUAL
327	MAR13	.	16955.12	899.71	15191.72	18718.52	.
328	APR13	.	13945.87	982.16	12020.87	15870.86	.
329	MAY13	.	11860.26	1026.83	9847.71	13872.80	.

This requires management to prepare an optimization of our facility to compensate for the change in demand in May 2013 There are five class of customers; high end restaurant, middle tier restaurant, fast food restaurant, grocery store, and pet food. The following table summarizes expected demand from these customers' on average per month.

Table 1c: Demand Expected in Future Month Three(MAY 2013)

Customer	Demand %	Net Amount expected at month three (lbs)
High end restaurant	5%	5930.1
Middle tier restaurant	10%	1186.03
Fast food restaurant	30%	3558.08
Grocery Store	25%	2965.07
Pet food	20%	2372.05
Total	100%	11860.26

Next management has to determine what produce grade each customer will buy. This is assumed data, based on previous purchase history with the customer shown in table 1D.

Table 1D: Expected Demand by Grade per Customer

Grade	Whole	Boneless and Skinned	Ground/Processed
High end restaurant	444.76	148.25	0
Middle tier restaurant	593.01	593.01	0
Fast food restaurant	355.81	2490.65	711.62
Grocery Store	889.52	1779.04	296.51
Pet food	0	830.22	1541.83

By importing our demand data into the Proc Optmodel that is set up to place the forecast demand into the supply function, an optimization schedule for the anticipated change in demand in three months can be determined to maximize our profit.

II. Profit Optimization Model

Before we can use the PROC OPTMODEL function in SAS to determine the optimized production schedule, one must analyze the production variables. The catfish processing and packaging company is a small firm, with four employees handling all of the processing and packaging of the fish. Each employee has a slightly different skill set and efficiencies for each grade of fish that is processed. Pay rates are not equal for the four employees. Due to these factors, the revenue from each pound of product processes by an employee also not equal. Table 1E, shows the revenue per pound of each grade of fish, per customer, for employees number 1 and 2. The full dataset for all four employees can be found in the SAS code located in the index. This data was generated by the group as for the sake of this project.

Table 1E: Revenue per Unit of Each Grade of Fish
Per Processor, Per Customer
(cents)

Processor	Customer	Whole	Boneless and Skinned	Ground/Processed
Employee 1	High end restaurant	125	178	0
Employee 1	Middle tier restaurant	125	178	0
Employee 1	Fast food restaurant	125	178	0
Employee 1	Grocery Store	125	178	0
Employee 1	Pet food factory	0	178	0
Employee 2	High end restaurant	125	168	0
Employee 2	Middle tier restaurant	125	168	0
Employee 2	Fast food restaurant	125	168	100
Employee 2	Grocery Store	125	168	100
Employee 2	Pet food factory	0	168	100

A major factor that plays a role not only in production scheduling, but profit, is an employee's production level. Table 1F displays the labor time (percentage based) an employee has spent processing each grade of fish. Also included is the maximum amount of product that employee can process in a given time period. As shown, not all employees spend the same amount of time processing each grade of catfish. This can be assumed to be based on skill level or seniority. As shown, Employee 1 spending the majority of his or her working day processing Boneless and Skinned catfish, but does not process the Ground/Processed grade. On the other hand, Employee 4 spends no time with the Boneless and Skinned grade, but spends the majority of his or her time at the grinder. Values given do not add up to 1.00, as we have to take into account the setup and cleanup time when transitioning from one grade to another or ending the shift.

Table 1F: Time Spent Processing Each Grade and Maximum Processing

Processor	**Whole**	**Boneless and Skinned**	**Ground/Processed**	**Maximum Capacity**
Employee 1	0.250	0.600	0.000	5000
Employee 2	0.250	0.500	0.215	4000
Employee 3	0.250	0.550	0.200	3000
Employee 4	0.250	0.000	0.600	5000

Using the SAS code found in the index, PROC OPTMODEL was run and returned the following results:

		grade		
		1	2	3
		amount	amount	amount
		Sum	Sum	Sum
processor	customer			
1	1	.	1482.53	.
	2	.	593.01	.
	3	.	2490.65	.
	4	.	1779.04	.
	5	.	830.22	.
4	1	4447.58	.	.
	2	593.01	.	.
	3	355.81	.	711.62
	4	889.52	.	296.51
	5	.	.	1541.83

As seen in the SAS output, only Employees 1 and 4 are listed as processors, meaning that our firm can meet demand as shown in Table 1D and maximize revenue without Employees 2 and 3. This model favors skill specialization over a group of employees that contribute in all or most areas. Employee 1 will be assigned to process all Boneless and Skinned catfish while Employee 4 will package all whole fish and ground/processed fish.

III. Conclusion and Recommendations

After using historical sales data to predict demand for catfish sales using an ARIMA model, the PROC OPTMODEL function in SAS Enterprise Guide was used to optimize a production schedule to maximize profit while meeting demand for our five customer groups with three grades of product. The results of this analysis indicate that our firm should make a major change in staffing. The current processing staff consists of 4 individuals, two of which work in specialized roles and two of which who contribute to processing all the catfish grades we sell. The optimization model suggests not only should we shift to a more specialized workforce, but that Employees 2 and 3 are not needed in the production schedule.

Our recommendation to senior management is to gather more data and analyze other potential opportunities before making a drastic change to staffing that would cut the production employee numbers in half. Issues such as cross training, and employee turnover must be considered before reducing the staffing level. What-if scenarios in which one employee is forced to complete all the processing for a period of time should be presented, should the staffing level be reduced to one through termination, medical leave, or vacation time.

Another opportunity for increased profit that should be analyzed is the price set by the firm. Currently, the firm charges the same price per pound, regardless of grade. What-if scenarios in which a different price is assigned to each grade of product, or even a different contract price for each customer should be evaluated. If price changes such as these would be beneficial to the firm, the optimization model should be reevaluated using the estimated revenue figures. This analysis should be completed before making drastic staffing changes as additional staff may be necessary to optimize the profit and production schedule.

IV. Index

Code for demand:

```
data fish;
set catsold;
RETAIN DATE '01Dec85'D;
DATE = INTNX('MONTH',DATE,1);
FORMAT DATE MONYY.;
t+1;
Price=F2;
Sales=F1;
run;

data future;
input t Sales Price;
RETAIN DATE '01FEB13'D;
DATE = INTNX('MONTH',DATE,1);
FORMAT DATE MONYY.;
datalines;
327 . 276.6
328 . 276.6
329 . 322
;
run;

data comb;
set fish future;
drop f1 f2;
run;

proc print data=comb;
run;

ods graphics on;
proc arima data=comb out=outarima;
identify var=Sales(1) crosscorr=Price(1) nlag=36;
estimate input=Price p=(1)(12) q=(1)(12) method=ml noconstant;
forecast lead=3 ID=Date interval=month;
run;

proc print;
run;
```

```
options ls = 70 ps = 59;
goptions device = PSLMONO reset = global
   gunit = pct ftext = swissb htitle = 3 htext = 2
   gsfmode = append hsize = 6.in vsize = 6.in;
filename gout1 'plot.ps';
goptions gsfname = gout1;

proc gplot data=fish;
   plot sales * date = 1;
   symbol1 v = 'none' i = spline;
   symbol2 v = triangle;
run;
proc gplot data=fish;
   plot price * date = 1;
   symbol1 v = 'none' i = spline;
   symbol2 v = triangle;
run;
```

Code for proc optmodel:

```
options ls = 70 ps = 59;

/* Revenue per unit of each grade of fish, per processor, per customer
*/

data object;
input processor customer grade1 grade2 grade3;
datalines;
1 1 125 178 .
1 2 125 178 .
1 3 125 178 .
1 4 125 178 .
1 5 . 178 .
2 1 125 168 .
2 2 125 168 .
2 3 125 168 100
2 4 125 168 100
2 5 . 168 100
3 1 125 167 .
3 2 125 167 .
3 3 125 167 100
3 4 125 167 100
3 5 . 167 100
4 1 125 . .
4 2 125 . .
4 3 125 . 100
4 4 125 . 100
4 5 . . 100
;
run;

/* Labor time capacities and consumption per unit of each grade */

data resource;
input processor
grade1 grade2 grade3 avail;
datalines;
1 .250 .600 . 5000
2 .250 .500 .215 4000
3 .250 .525 .200 3000
4 .250 . .600 5000
;
run;

/* Demand by each customer for each grade of fish, where grade
1=Whole, Grade 2=Boneless/Skinned and Grade 3=Ground/processed*/

data demand;
input customer grade1 grade2 grade3;
```

```
datalines;
1 4447.58 1482.53 0
2 593.01 593.01 0
3 355.81 2490.65 711.62
4 889.52 1779.04 296.51
5 0 830.22 1541.83
;
run;



/* Demand by each customer for each grade of fish */

proc optmodel printlevel=0;
set CUSTOMERS;
set GRADES = 1..3;
set PROCESSORS;

/* parameters */

number profit{CUSTOMERS, GRADES, PROCESSORS} init 0;
number demand{CUSTOMERS, GRADES};
number time{GRADES,PROCESSORS} init 0;
number avail{PROCESSORS} init 0;

/* load the customer set and demands */

read data demand
into CUSTOMERS=[customer]
{g in GRADES} <demand[customer,g]=col("grade"||g)>;

/* load the processor set, time costs, and availability */

read data resource nomiss
into PROCESSORS=[processor]
{g in GRADES} <time[g,processor]=col("grade"||g)> avail;

/* load objective data */

read data object nomiss
into [processor customer]
{g in GRADES} <profit[customer,g,processor]=col("grade"||g)>;

/* the model */

var y{CUSTOMERS, GRADES, PROCESSORS} >= 0;
max obj = sum{c in CUSTOMERS, g in GRADES, m in PROCESSORS}
    profit[c,g,m]*y[c,g,m];
con req_demand{c in CUSTOMERS, g in GRADES}:
sum{m in PROCESSORS} y[c,g,m] = demand[c,g];
con req_avail{m in PROCESSORS}:
```

```
sum{c in CUSTOMERS, g in GRADES} time[g,m]*y[c,g,m] <= avail[m];

/* call the solver and save the results */

solve with lp/solver=primal;
create data solution
from [customer grade processor]
={c in CUSTOMERS, g in GRADES, m in PROCESSORS: y[c,g,m]^=0}
amount=y;
quit;


proc print data=solution;
proc tabulate data=solution;
class customer grade processor;
var amount;
table (processor*customer), (grade*amount);
run;
```